

# EESy Solutions

## Engineering Equation Solver Newsletter

### Inside this issue:

Welcome	1
Equation Window Tabs	1
The \$InsertTab Directive	3
Example: Using Tabs for Procedures	3
Example: Tabs w/Different EES Programs	6
Adjusting Diagram Window Attributes	8
Plotting Directives	9
Recent Changes to EES	11
Instant Update & Technical Service	11


### Welcome

This is the 45<sup>th</sup> issue of EESy Solutions, a newsletter that provides news, tips, and other updates for users of the Engineering Equation Solver software. This issue discusses the introduction of Equation Window Tabs, a new feature available in the EES Professional license. These tabs can contain code segments e.g., Functions and Procedures, or they can contain completely separate and independent EES programs. The use of tabs is a powerful and convenient way to organize EES codes. This issue also presents several directives that have been added to the Professional license to automate the generation of plots based on the solution of an EES program.

EES has been commercially available for more than two decades. Previous issues of EESy Solutions, they can be downloaded from <https://fchartsoftware.com>.

### Equation Window Tabs

Starting with version 11.695, the Professional license allows the equations in an EES program to be organized into tabs. Prior to v11.695, all equations had to be placed into one window (the Equations Window), as shown in the figure below on the left. Tabs allow the user to organize the equations into sub-windows, as shown in the figure below on the right. The equations in the tabs can be separate EES programs or they can be connected using the \$InsertTab directive.




(Left) The EES program before v11.695 with all equations in a single Equations Window. (Right) Professional license of EES after v11.695 allows the Equations Window to be organized into different tabs.

## Equations Window Tabs (cont.)


The different tabs can hold code segments like Functions, Procedures, Subprograms, and Modules that you would like to use in the code contained in the other tabs. Alternatively, the tabs can hold independent and stand alone, but related, EES programs that you would like to store in a single EES file.

In order to add a tab, right click on an existing tab and select Add Tab. Then enter the desired Tab name in the resulting dialog and your Equations Window will show a new tab.




To add a tab (Left) Right click on an existing tab and select Add Tab, (Center) enter the name of the tab and (Right) the new tab should appear.

By default the tabs operate independently of one another. For example, in the Main tab we can enter a complete set of equations and in Tab 1, a different and also complete set of equations. Clicking on the Main Equation Tab and solving (F2) leads to the solution for one set of equations, whereas clicking Tab 1 and solving leads to the solution to the other set.




(Left) Equations entered in Main Tab and (Right) equations entered in Tab 1



(Left) Solution to equation entered in Main Tab and (Right) in Tab 1

By default, the equations in each tab are independent of one another. If a set of equations is entered in the Main Tab that relies on equations from Tab 1 then it will not be possible to obtain a solution to the equations in the Main Tab unless they are linked together.




(Left) Equations entered in Main Tab that depend on (Right) equations entered in Tab 1

Error message obtained when trying to solve the equations in the Main Equations Tab.

## The \$InsertTab Directive


Equations in different tabs can be linked using the \$InsertTab directive. For example, if the \$InsertTab directive is placed in the Main Equations Tab then the equations from Tab 1 are inserted into the Main Equation at the location of the \$InsertTab directive.



(Left) The \$InsertTab directive in the Main Equations Tab is used to insert the equations contained in Tab 1 (Middle) to get the solution to the coupled set of equations (Right).

Because there is no \$InsertTab directive in Tab 1, solving the equations in Tab 1 will only solve those equations.


The use of tabs together with the \$InsertTab directive allows an EES program to be carefully organized in various ways. For example, you can place code segments that will be used in several EES programs (e.g., Functions and Procedures) in separate tabs and then include those code segments in other tabs with the \$InsertTab directive. You can also place independent, but related, EES codes in different tabs so that they are saved together and reference the same set of common code segments.



(Left) Equations entered in Tab 1 and (Right) solution to the equations entered in Tab 1

## Example: Using Tabs for Procedures

Note: This example and the two graphic files it requires can be downloaded from the [EESySolns page](#) of our website. Consider the vapor compression cycle that uses a liquid-suction heat exchanger shown below.




Vapor compression cycle with a suction-liquid heat exchanger.

## Example: Using Tabs for Procedures (cont.)

The cycle uses refrigerant R1234yf and the operating conditions for the cycle are  $T_C = -5^\circ\text{C}$  and  $T_H = 25^\circ\text{C}$ . The compressor has a displacement rate of 70 cc/s, a volumetric efficiency of 78% and an isentropic efficiency of 72%. The superheat and subcool are  $3^\circ\text{C}$  and  $5^\circ\text{C}$ , respectively. The approach temperature differences for the condenser and evaporator are  $5^\circ\text{C}$  and  $4^\circ\text{C}$ , respectively. The suction-liquid heat exchanger effectiveness is 70%. The cycle consists of a compressor, an expansion valve, a recuperative heat exchanger and two additional heat exchangers. The Procedures used to simulate some of these components are placed in separate tabs in order to organize the code and also allow use of these code segments from other tabs.

For example, let's define a new tab called Compressor and enter a Procedure to simulate the compressor within this tab. This Procedure takes in the inlet temperature and pressure, exit pressure, refrigerant, displacement rate, and the volumetric and isentropic efficiencies. It returns the outlet enthalpy, entropy, and temperature as well as the mass flow rate and power. We can also define a new tab called Recuperator and enter a Procedure to simulate the suction-liquid heat exchanger within this tab. This Procedure takes in the inlet temperatures and pressures on the hot and cold sides as well as the effectiveness and provides the outlet conditions (enthalpy, entropy, and temperature) on both sides.



```


Equations Window
Main Compressor Recuperator
Procedure compressor(T_in, P_in, P_out, R$, V_dot_disp, eta_vol, eta_c: h_out, s_out, T_out, m_dot, W_dot)

    "Inlet conditions"
    h_in=enthalpy(R$, T=T_in, P=P_in)
    s_in=entropy(R$, T=T_in, P=P_in)
    v_in=volume(R$, T=T_in, P=P_in)
    h_out_s=enthalpy(R$, P=P_out, s=s_in)
    h_out=h_in+(h_out_s-h_in)/eta_c
    "isentropic outlet enthalpy"
    "actual outlet enthalpy"

    "outlet conditions"
    T_out=temperature(R$, h=h_out, P=P_out)
    s_out=entropy(R$, h=h_out, P=P_out)

    "mass flow rate and power"
    m_dot=V_dot_disp*eta_vol/v_in
    W_dot = m_dot*(h_out-h_in)
End
  
```

*Compressor procedure is placed in the Compressor Tab*



```

Equations Window
Main Compressor Recuperator
Procedure recuperator(R$, T_H_in, P_H, T_C_in, P_C, eff: h_H_out, s_H_out, T_H_out, h_C_out, s_C_out, T_C_out)

    h_H_in = enthalpy(R$, T=T_H_in, P=P_H)
    h_C_in = enthalpy(R$, T=T_C_in, P=P_C)
    h_H_out_min=enthalpy(R$, T=T_C_in, P=P_H)
    Q_dot_H_max=(h_H_in - h_H_out_min)
    h_C_out_max=enthalpy(R$, T=T_H_in, P=P_C)
    Q_dot_C_max=(h_C_out_max - h_H_in)
    Q_dot=eff*min(Q_dot_H_max, Q_dot_C_max)
    h_H_out=h_H_in-Q_dot
    s_H_out=entropy(R$, h=h_H_out, P=P_H)
    T_H_out=temperature(R$, h=h_H_out, P=P_H)
    h_C_out=h_C_in+Q_dot
    s_C_out=entropy(R$, h=h_C_out, P=P_C)
    T_C_out=temperature(R$, h=h_C_out, P=P_C)
End
  
```

*Recuperator procedure is in the Recuperator Tab*

These two code segments (Compressor and Recuperator) can be accessed from any other tab using the \$InsertTab directive.

## Example: Using Tabs for Procedures (cont.)

The Main Equations Window Tab is used to simulate the vapor compression cycle with the suction-liquid heat exchanger. Note the \$InsertTab statements must be placed at the top because Procedures must be entered before any of the equations in the main program appear.


Main	Compressor	Recuperator
<pre> \$UnitSystem SI Mass J Pa K \$InsertTab Compressor \$InsertTab Recuperator  R\$='R1234yf' T_H = converttemp(C,K,25 [C]) T_C = converttemp(C,K,-5 [C]) eta_c = 0.72 [-] eta_vol = 0.78 [-] V_dot_disp = 70 [cm^3/s]*convert(cm^3/s,m^3/s) DT_sc = 5 [K] DT_sh = 3 [K] DT_evap = 4 [K] DT_cond = 5 [K] eff_sl = 0.7 [-]  "condenser exit" T[4]=T_H+DT_cond P[4]=pressure(R\$,x=0,T=T[4]+DT_sc) h[4]=enthalpy(R\$,T=T[4],P=P[4]) s[4]=entropy(R\$,T=T[4],P=P[4])  "evaporator exit" T[1]=T_C-DT_evap P[1]=pressure(R\$,x=1,T=T[1]-DT_sh) h[1]=enthalpy(R\$,T=T[1],P=P[1]) s[1]=entropy(R\$,T=T[1],P=P[1])  "suction-liquid heat exchanger" Call recuperator(R\$, T[4], P[4], T[1], P[1], eff_sl: h[5], s[5], T[5], h[2], s[2], T[2]) P[2]=P[1] P[5] = P[4]  "throttle exit" P[6]=P[1] h[6]=h[5] T[6]=temperature(R\$,P=P[6],h=h[6]) s[6]=entropy(R\$,P=P[6],h=h[6])  "Compressor" P[3] = P[4] Call compressor(T[2], P[2], P[3], R\$, V_dot_disp, eta_vol, eta_c: h[3], s[3], T[3], m_dot, W_dot_c)  Q_dot_evap=m_dot*(h[1]-h[6]) COP=Q_dot_evap/W_dot_c </pre>		
	<pre> "refrigerant" "ambient temperature" "cold space temperature" "isentropic efficiency of compressor" "volumetric efficiency of compressor" "displacement rate of compressor" "degree of sub-cooling" "degree of super-heating" "evaporator approach" "condenser approach" "SL heat exchanger effectiveness" </pre>	

*Simulation in Main Equations Tab for the cycle with the suction-liquid heat exchanger.*




## Example: Tabs with Different EES Programs

We can also put independent EES programs in different tabs. For example, we might want to simulate different cycles using the same inputs. In this case we can define a new tab called Inputs in which the inlet conditions and other parameters are entered. The Inputs Tab can be included in any other tab that provides a similar refrigeration cycle simulation.



*Inputs Tab contains inputs that are common to multiple cycle simulations*

The previously developed simulation for the vapor compression cycle with a suction-liquid heat exchanger in the Main Equations Window Tab is modified so that the inputs are obtained from the Inputs Tab.




*Vapor compression cycle with liquid-suction heat exchanger modified to get input parameters from the Inputs Tab.*

## Example: Tabs with Different EES Programs

A second simulation can be developed for a vapor compression cycle that does not include a suction liquid heat exchanger. The equations have been entered in a new tab named Basic VC Cycle.

The Basic VC Cycle uses the \$InsertTab directive to include the code segments for the compressor (in the Compressor Tab) and also the inputs (in the Inputs Tab). The simulation itself is different from and independent of the simulation contained in the Main Equations Tab but it uses some of the same code segments.



*Vapor compression cycle without a suction-liquid heat exchanger.*

```

x=1 Equations Window
Main Inputs Compressor Recuperator Basic VC Cycle
$UnitSystem SI Mass J K Pa
$InsertTab Compressor
$InsertTab Inputs

"condenser exit"
T[4]=T_H+DT_cond
P[4]=pressure(R$,x=0,T=T[4]+DT_sc)
h[4]=enthalpy(R$,T=T[4],P=P[4])
s[4]=entropy(R$,T=T[4],P=P[4])

"evaporator exit"
T[2]=T_C-DT_evap
P[2]=pressure(R$,x=1,T=T[2]-DT_sh)
h[2]=enthalpy(R$,T=T[2],P=P[2])
s[2]=entropy(R$,T=T[2],P=P[2])

"throttle exit"
P[1]=P[2]
h[1]=h[4]
T[1]=temperature(R$,P=P[1],h=h[1])
s[1]=entropy(R$,P=P[1],h=h[1])

"Compressor"
P[3]=P[4]
Call compressor(T[2], P[2], P[3], R$, V_dot_disp, eta_vol, eta_c: h[3], s[3], T[3], m_dot, W_dot_c)

Q_dot_evap=m_dot*(h[2]-h[1])
COP=Q_dot_evap/W_dot_c
  
```

B|W US Line Numbers: Off Wrap: On Insert Caps Lock: Off SI K Pa J mass deg Warnings: 1


*The Basic VC Cycle Tab contains the simulation of a vapor compression cycle without a suction-liquid heat exchanger.*

## Adjusting the Diagram Window Attributes

Only one Diagram Window (with its children) is available for the entire EES program. However, the attributes in the Diagram Window can be adjusted programmatically in order to adapt to the code in different tabs. For example, the cycle schematic shown in the Diagram Window can be changed based on the tab for which equations are being solved using the File-Name\$ attribute of the picture object.



Open the Diagram Window and put it in Application mode (if it is not already in that mode); this can be done by selecting Ctrl-D until the A appears in the upper right corner and the Diagram Window toolbar appears. Select the Add Picture from File button from the toolbar and provide a name for the picture object (e.g., CycleImage). Click on the File-Name\$ attribute box to bring up the Select Picture File dialog. Navigate to the image you would like to have displayed in the Diagram Window.

add picture  
from file



(Left) Diagram Window in Application Mode with the Add Picture from File tool highlighted and (Right) Select Picture File dialog with the picture object named CycleImage.

The image can be changed depending on which Equations Tab is being executed by setting the File-Name\$ attribute of the CycleImage object with an equation from within the tab, as shown below.



The Diagram Window when the (Left) Basic VC Cycle Tab is executed and (Right) the Main Equations Tab is executed.



## Plotting Directives


A number of new plotting directives are available with the Professional license to automate the creation of plots based on the results of an EES model. Information for all directives can be found by selecting Directive Info from the Options menu to access the Directive Information dialog. The new directives include \$PropertyPlot, \$NewPlot, and \$OverlayPlot. The \$PropertyPlot creates a new property plot. The arguments to the \$PropertyPlot directive include the fluid type, plot type, number and values of the constant property lines (e.g., isobars and isochors for a T-s diagram), temperature range, and name.

For example, the directive below placed at the bottom of the Main Equations Tab:

`$PropertyPlot R$ TS 2 P[2] P[3] 0 TMax = 400 TMin = 240 Name = 'TsPlot'`

creates a Temperature-Entropy plot for the refrigerant being used (R\$) with 2 isobars that have values associated with P[2] and P[3] after calculations are completed. The minimum and maximum temperatures used to construct the plot are 240 K and 400 K. The plot is named TsPlot. If a plot named TsPlot already exists, then the \$PropertyPlot directive will delete that plot and make a new one.

The \$OverlayPlot directive can be used to overlay the current simulation state points onto the property plot. The arguments of the \$OverlayPlot directive include the name of the plot to be used as the basis of the overlay, the source of the data, and information about the symbol size, and color to be used. Optionally, the points on the plot can be labeled based on their subscript.



*The Directive Information dialog showing directives related to plotting.*

For example, the directive below placed immediately below the \$PropertyPlot directive in the Main Equations Tab will overlay the current state points onto the T-s plot and identify the state points.

`$OverlayPlot Name='TsPlot' Table=ARR1 X=s[i] Y=T[i] Rows=1..6 Line=0 Symbol=9 SymbolSize=6 Color=Red PointLabels`

The use of the \$PropertyPlot and \$OverlayPlot directives will cause the plot TsPlot to be deleted and re-made each time the equations in the Main Equations tab are executed. The resulting T-s plot that is created will be customized to the current simulation of the vapor compression cycle with a suction liquid heat exchanger.

A similar set of directives in the Basic VC Cycle Tab will cause the TsPlot to be updated based on the results of that simulation allowing the plot to adjust to the equations in each tab.

## Plotting Directives (cont.)

```

Equations Window
Main Inputs Compressor Recuperator Basic VC Cycle

Q_dot_evap=m_dot*(h[2]-h[1])
COP=Q_dot_evap/W_dot_c

$PropertyPlot R$ TS 2 P[2] P[3] 0 TMax = 400 TMin = 240 Name = 'TsPlot'
$OverlayPlot Name='TsPlot' Table=ARR1 X=s[i] Y=T[i] Rows=1..4 Line=0 Symbol=9 SymbolSize=6 Color=Red PointLabels
  
```

B|W US Line Numbers: Off Wrap: On Insert Caps Lock: Off SI K Pa J mass deg Warnings: On Unit Chk: Auto Complex:

```



Equations Window
Main Inputs Compressor Recuperator Basic VC Cycle

Q_dot_evap=m_dot*(h[1]-h[6])
COP=Q_dot_evap/W_dot_c

$PropertyPlot R$ TS 2 P[2] P[3] 0 TMax = 400 TMin = 240 Name = 'TsPlot'
$OverlayPlot Name='TsPlot' Table=ARR1 X=s[i] Y=T[i] Rows=1..6 Line=0 Symbol=9 SymbolSize=6 Color=Red PointLabels
  
```

B|W US Line Numbers: Off Wrap: On Insert Caps Lock: Off SI K Pa J mass deg Warnings: On Unit Chk: Auto Complex:

The \$PropertyPlot and \$OverlayPlot directives placed in the (Top) Basic VC Cycle Tab and (Bottom) Main Equations Tab.



The T-s Plot generated when the (Left) Basic VC Cycle Tab and (Right) Main Equations Tab are executed.

## Recent Changes to EES

- Tabs can be added to the Equations Window with the Professional EES license and interfaced using the \$InsertTab directive.
- The \$OverlayPlot, \$NewPlot, and \$PropertyPlot directives are available in the Professional license to facilitate the generation of plots with commands in the Equations Window.
- Property data for Midel-7131, R471a, R1336mzz(E), and R424a have been added.
- Uncertainty information can be displayed using either error bars or as a shaded region.
- The \$VarInfo directive now allows a variable to be designated with a Hidden Style, causing it to be hidden in the Solution Window.
- Thermodynamic information for R11, R12, R13, R14, R114, R141b, R508B, and R1336mzz(Z) have been updated.
- The \$ShowWindow directive coupled with the GetWindowInfo command in the Professional version allows the position and size of each EES window to be specified.
- Customized unit lists for basic and alternate units can be saved in .unt1 and .unt2 files.
- The capabilities of the Variable Information Palette that appears to the right of the Main Equations Window have been updated. Left clicking on a variable with the Ctrl key pressed will insert the variable. Right clicking will search for the variable.
- The NBS Metal Property library provides the conductivity of several metals as a function of temperature and residual resistivity ratio (RRR).
- The Professional license allows text items and graphic objects in the Diagram Window to be displayed before calculations, after calculations, or all of the time.
- The Substance Information dialog can be accessed from the Options menu and allows a systematic search of the more than 2700 substances that are currently available in EES.
- The Replace\$ string function allows a substring to be replaced by a different substring within a string.
- Transport properties for the substances CarbonDioxide, Propane, and Isobutane have been improved.



**F-Chart Software**

PO Box 44042  
Madison, WI, 53744

Phone: 608-274-4262  
Internet: <http://fchartsoftware.com>  
E-mail: [info@fchart.com](mailto:info@fchart.com)

## Instant Update & Technical Service

EES uses a different model for updating than most other programs. Each time that there is a change in the EES program, either to correct a problem or to add a new feature, the version number is incremented by 0.001 and the latest version of EES is placed on our website. Although the program has become very robust and stable, there have been many new versions of EES released since the last EESy Solutions was distributed.

Any user who has a current subscription to our Instant Update & Technical Service (IUTS) or Academic Update Service (AUS) can download the current version. All new Commercial and Professional licenses of EES are provided with a one year subscription of this service. The cost to continue IUTS or AUS after the first year is ~20% of the current cost of the program per year, provided that it is renewed within 12 months after expiration. Contact us if you wish to re-subscribe to IUTS or AUS.